
*
*   VIN      R1                                           R2
*     _/\  /\  /\_ OUT0                         OUT1 __/\  /\  /\__
*    |   \/  \/   |         IABC= 1mA->0            |    \/  \/   _|_
*   _|_           |                                 |             \\\
*  /_  \          |        INP0=INN1    INP1=INN0   |
* // \  \     /|  |      /|________    _____|\      |  |\  VOUT1
* \   \//  __/ |__|_/\/\/ |        |  |     | \/\/\_|__| \__
*  \___/  |  \ |    \/\/\ |_______/|\_|   __| /\/\/    | /  |
*   _|_   |   \|         \| |      |_____|  |/         |/   |
*   ///   |B_BUF0  B_OTA0   |            | B_OTA1           |
*         |           R3    |   R4       |       R5         |
*         |_____/\  /\  /\__|_/\  /\  /\_|____/\  /\  /\____|
*       VOUT0     \/  \/  1NN0  \/  \/  INN1    \/  \/
*
*
*
                    

VIN       VIN        0     SIN(   0    10m   1000   )  
R1        VIN        OUT0  10K
R2        OUT1       0     10K
R3        VOUT0      INN0  100K
R4        INN1       INN0  1K
R5        VOUT1      INN1  100K
B_OTA0    OUT0       0     I =   -1*v(VIABC)*tanh(( v(INN1) -v(INN0) )/.052)
B_BUF0    VOUT0      0     V =    v(OUT0) -1.2
B_OTA1    OUT1       0     I =   -1*v(VIABC)*tanh(( v(INN0) -v(INN1) )/.052)
B_BUF1    VOUT1      0     V =    v(OUT1) -1.2
V_Iabc    VIABC      0     PWL    ( 0 1m 10m 0 ) 
.tran     1u         10m   0      1u   

.control
run
set       pensize = 2
plot      v(vin)    v(out0) v(out1)
.endc

.end

=====================END_OF_SPICE============================

This code works with winspice.

This example shows how the floating resistor shown in the
data sheet operates. In this case the actual 1.2V offset of the
buffers can be added. All input voltages to the OTAs will float 
to be 1.2V lower in voltage compared to nodes OUT0 and OUT1.  



*
*   VIN      R1                                           R2
*     _/\  /\  /\_ OUT0                         OUT1 __/\  /\  /\__
*    |   \/  \/   |         IABC= 1mA->0            |    \/  \/   _|_
*   _|_           |                                 |             \\\
*  /_  \          |        INP0=INN1    INP1=INN0   |
* // \  \     /|  |      /|________    _____|\      |  |\  VOUT1
* \   \//  __/ |__|_/\/\/ |        |  |     | \/\/\_|__| \__
*  \___/  |  \ |    \/\/\ |_______/|\_|   __| /\/\/    | /  |
*   _|_   |   \|         \| |      |_____|  |/         |/   |
*   ///   |B_BUF0  B_OTA0   |            | B_OTA1           |
*         |           R3    |   R4       |       R5         |
*         |_____/\  /\  /\__|_/\  /\  /\_|____/\  /\  /\____|
*       VOUT0     \/  \/  1NN0  \/  \/  INN1    \/  \/
*
*
*
B_OTA0    OUT0    0     I =   -1*v(VIABC)*tanh(( v(INN1) -v(INN0) )/.052)
B_BUF0    VOUT0   0     V =    v(OUT0)      -1.2
B_OTA1    OUT1    0     I =   -1*v(VIABC)*tanh(( v(INN0) -v(INN1) )/.052)
B_BUF1    VOUT1   0     V =    v(OUT1)      -1.2
V_Iabc    VIABC   0     PWL    ( 0 1m 10m 0 ) 

The floating resistor can be put in a symmetrical network
to view its behavior on both of its nodes as IABC moves from 1mA to zero. 

It looks like at 1mA the equivalent input resistance
is about 10K Ohms. As IABC decreases, Node OUT0 approaches
VIN and Node OUT1 approaches ground. 

*
*   VIN      R1                                           R2
*     _/\  /\  /\_ OUT0                         OUT1 __/\  /\  /\__
*    |   \/  \/   |         IABC= 1mA->0            |    \/  \/   _|_
*   _|_           |                                 |             \\\
*  /_  \          |        INP0=INN1    INP1=INN0   |
* // \  \     /|  |      /|________    _____|\      |  |\  VOUT1
* \   \//  __/ |__|_/\/\/ |        |  |     | \/\/\_|__| \__
*  \___/  |  \ |    \/\/\ |_______/|\_|   __| /\/\/    | /  |
*   _|_   |   \|         \| |      |_____|  |/         |/   |
*   ///   |B_BUF0  B_OTA0   |            | B_OTA1           |
*         |           R3    |   R4       |       R5         |
*         |_____/\  /\  /\__|_/\  /\  /\_|____/\  /\  /\____|
*       VOUT0     \/  \/  1NN0  \/  \/  INN1    \/  \/
*
*
*
To make a floating inductor, a capacitor can be placed
across R4. To make a floating capacitor, matched
capacitors need to be placed across R3 and R5.  


